Modeling Lightning Arresters Using Spice

F.J. Martínez Esteban, L.C. Montañés Bellosta, J. Mur Amada,

A. Llombart Estopiñan, M.García Gracia.

Departamento de Ingeniería Eléctrica, Electrónica y Comunicaciones.

Area de Ingeniería Eléctrica.

Universidad de Zaragoza.

ABSTRACT

It is difficult to understand the lightning performance of a transmission line during a lightning stroke without using a simulation program. SPICE was selected as the software to simulate this lightning phenomenon on electric power transmission lines. Instead of the limitation of the non-linear elements libraries in SPICE, we have attempt to simulate a lightning arrester with this program.

The models developed to simulate the insulator strings and lightning arresters will be presented. Some application examples to illustrate the performance of this model are also discussed.

1. INTRODUCTION

A lightning stroke is a source which injects an exponential current into a transmission line at the point of contact. Since both the lightning stroke and the transmission line exhibit linear characteristics, the modelling of these parts of the simulation is straightforward. In contrast, the insulators and lightning arresters are devices which exhibit non-linear characteristics. They are designed to have a very high impedance for nominal operating voltage; however, if the voltage on the transmission system exceeds design specifications, these devices are designed to sparkover, that is, allow an arc to form. This arc is a low impedance path to ground which is the mechanism used to remove the excess energy from the line before it can damage valuable equipment.

All of the components used to simulate lightning phenomenon are available in PSPICE [1]; by this reason, L. A. Kraft [2] employed this software. But this model doesn't give the expected results in some cases. The present model is capable to predict the electrical performance of a lightning arrester in power distribution systems.

2. MODELLING INSULATORS AND LIGHTNING ARRESTERS

All lightning strokes exhibit the basic characteristic of an exponential current source that is defined by:

$$i(t) = -I_0 \cdot \left[e^{-t/\tau_1} - e^{-t/\tau_2} \right] \tag{1}$$

The magnitude of the current waveform I_O, and the rise and fall constants (τ_1 and τ_2) will vary with every lightning stroke, but they can be studied statistically. Therefore, the lightning stroke can be modelled in SPICE [3] by an exponential current source.

High voltage power cables have traditionally designed combining empirical rules, based on experimental evidence, with the solution of electromagnetic and thermal fields using finite difference approaches [4,5]. In the cases presented below, the transmission line model included in the software package is used.

The modelling of the insulator strings on the transmission towers and lightning arresters involves the modelling of non-linear devices. The developed model for the lightning arrester consists on two resistors in parallel: the first one (R_1) has low impedance, and the other (R_2) has a high impedance. The branch with the low resistance has a voltage switch (*vswitch*) in series (see Figure 1).



Figure 1: General model for both the insulator string and lightning arrester.

When the voltage on the transmission line exceeds the design voltage, the *vswitch* turns on, and the impedance of the lightning arrester becomes very low, thus allowing the energy contained in the lightning to be removed from the line. In other case, the *vswitch* is off, and the impedance of the circuit turns very high.

The voltage controlled sources can only be adjusted by polynomial functions in SPICE. Our *vswitch must* work like a step voltage, which is impossible to simulate by a polynomial function due to the nature of the step function [6]. The design of the *vswitch* by using SPICE permits us to model the lightning arrester. It can be achieved with the block diagram showed in Figure 2.



Figure 2: Diagram block of the developed model.

In the figure 2, the inversion block transforms the negative voltage of the lightning stroke into a positive one to be compared with a reference.



Figure 3: Hysteresis of the comparator.

On the other hand, the comparator block consists on a hysteresis comparator (Figure 3) that is *high* when there is no lightning stroke (the input voltage is lower than a reference), and it switch off when it appears. Then, the comparator is *low* until the current is bellow a minimum.

Finally, the output block multiplies the input voltage by the logic output of the comparator, and it prompts R_1 to be connected to ground or to the input voltage (so that any current flows though it).



Figure 4: Circuit of the lightning arrester model developed.

```
.SUBCKT INSULATO 2 10
* Beginning of subcircuit LM111
R101 10 115 10K
R102 10 116 10K
D101 115 116 D101
D102 116 115 D102
Q101 115 7 111 NPN
Q102 116 5 111 NPN
IEE 111 12 400UA
GA 109 0 115 116 25.8MMHOS
RA 109 0 100K
D103 109 113 D103
VC 113 0 DC 30
D105 109 110 DIODE
RB 110 0 67K
GB 110 0 115 116 39MMHO
D104 114 109 D104
VE 0 114 DC 2.39
GC 13 117 109 0 52MMHO
IOS 13 117 126MA
RO 13 117 20
D106 6 117 DIODE
RPC 10 0 3.2K
RPE 12 0 4.1K
.MODEL D101 D(IS=8E-16 N=.386)
.MODEL D102 D(IS=8E-16 N=.262)
.MODEL NPN NPN(IS=8E-16)
.MODEL DIODE D(N=1M)
.MODEL D103 D(N=1M)
.MODEL D104 D(N=1M)
* End of LM111

* Beginning of subcircuit LN:
* calculates the half of the napierian
* logarithm of the voltage V4
D201 202 0 DIODEB
.MODEL DIODEB D(IS=1)
I201 202 0 1
E201 5 0 202 0 19.3317526509
G201 0 202 4 0 1
* End of subcircuit LN

* PRINCIPAL CIRCUIT
R10 2 3 "R10"
R21 7 8 "R21"
E14 11 0 10 0 "E14"

R9 0 2 100MEG
E11 4 0 0 2 0 1
R17 4 0 1K
R20 7 11 10K
R22 10 6 1K
E12 8 0 6 0 1
E15 12 0 10 0 -1
E16 13 0 10 0 -22.2912E-3
E10 3 0 POLY(2) 2 0 6 0 0 0 0 0 0 0.06667
.ENDS
```

Figure 5: List of the subcircuit *insulato*. The value of R10, R21 and E14 depend on the model of the insulator wanted to simulate.

The circuit of the model developed is shown in Figure 4, and the software corresponding is shown in Figure 5. The value of R_{10} represents the sparkover resistance of the device, while R_{21} and E_{14} are related with R_{10} and with the flashover voltage (V_{REF}) as follow:

$$R_{21} = 10^4 \cdot \left[\frac{30}{Ln(V_{REF}) - Ln\left(\frac{V_{REF} \cdot R_{10}}{Z_{LINE}}\right)} - 1 \right] \quad (2)$$

$$E_{14} = \frac{R_{21} + 10^4}{30 \cdot R_{21}} \cdot Ln\left(\frac{V_{REF} \cdot R_{10}}{Z_{LINE}}\right) \quad (3)$$

3. RESULTS

3.1. COMPARISON WITH ANOTHER MODELS.

This example illustrates the fall of a lightning just in the middle of a power line. One extreme is open and the other one is connected to a transformer (10 MVA, 20 KV), with equivalent capacitor to ground of the windings C=6nF. The lightning stroke is modelled by a pulse current source of 20 kA and the lightning stroke has a breakdown voltage of 400 V and an internal resistance of 4.5 Ω.



Figure 6: Circuit of this example.

```
*SPICE_NET
.TRAN 0.1U 40U
.OPTIONS ITL1=1E5 LIMPTS=500 ITL4=100
.OPTIONS ITL5=0 RELTOL=1E-3 ABSTOL=1E6
*INCLUDE DEVICE.LIB
*INCLUDE AVALVULA.LIB
*INCLUDE OWN.LIB
.OPTIONS TRTOL=2
*ALIAS  V(12)=VTRAFO
*ALIAS  V(1)=VIN
*ALIAS  V(11)=VEXTR
*ALIAS  I(V1)=IVAL
.PRINT TRAN  V(12)  V(1)  V(11)  I(V1)
T4 1 0 11 0 Z0=300 TD=16.66U
C2 12 0 6N
I2 0 1 PULSE 0 -20000 0 0 0 20U 150U
V1 12 18 0
V2 19 0 DC 15
X4 18 19 ARRESTER
T3 12 0 1 0 Z0=300 TD=16.66U
.END
```

Figure 7: Parameters employed for the simulation.

The results are compared with the theoretical values, and with the results obtained by means the model proposed by Kraft in an interesting work [2]. The parameters used for the simulation of the circuit proposed are shown in Figure 7.



Figure 8: Current absorbed by the lightning arrester.



Figure 9: Voltage at the point of contact of the lightning arrester.

By using the model of Kraft, we found oscillations which can be seen in Figures 8 and 9. It is owed to the presence of a inductor, which is required for a correct running. Voltage and current of the ideal model are superposed to the ones obtained with the model proposed above.

3.2. INSULATOR FLASHOVER

This example illustrates the performance of a transmission system when the current of the lightning stroke is large enough to cause flashover of the tower insulator strings.



Figure 10: Circuit used for the example.

The lightning stroke, with a crest current of approximately 25 kA, strikes a 138 kV transmission line with a surge impedance of 300 Ω, at midispan (see Figure 10).

```
*SPICE_NET
*INCLUDE DEVICE.LIB
.TRAN 0.02U 8U 0.000001F
.OPTIONS ITL1=1E5 ITL4=100 LIMPTS=1E3
.OPTIONS ITL5=0 RELTOL=1E-3 ABSTOL=1E2
*OWN.LIB has the model of the insulator.
*   The subcircuit INSULATOR has:
*      R10=20   R21=100.78K   E14=0.42184
*INCLUDE OWN.LIB
*INCLUDE NONLIN.LIB
.PRINT TRAN  I(V2) V(6)  I(V4) V(9)
.PRINT TRAN  I(V10) I(V15) I(V16) V(3)
.PRINT TRAN  I(V18) I(V19)
T4 2 0 3 0 Z0=300 F=60 NL=500U
T1 3 0 5 0 Z0=300 F=60 NL=100U
T2 6 0 8 0 Z0=300 F=60 NL=200U
T3 9 0 11 0 Z0=300 F=60 NL=200U
RL3 12 0 300
I2 0 4 EXP(0 -25K 0 1U 1.5U 5U)
V2 5 6 DC 0
V4 8 9 DC 0
V10 4 3 DC 0
V15 2 1 DC 0
V16 11 12 DC 0
V17 14 0 DC 15
V18 6 7 DC 0
X2 7 14 INSULATO
X3 10 13 INSULATO
V19 9 10 DC 0
V20 13 0 DC 15
RL4 1 0 300
.END
```

Figure 11: Parameters employed in the simulation of this example.

The transmission line has steel towers with 10 porcelain insulators with a sparkover voltage of 1.6 MegV. The tower surge resistance is 20 Ω.



Figure 12: Voltage at the point of stroke and at the insulators.



Figure 13: Current absorbed by the insulators.

The parameters employed in the simulation of the insulator flashover are shown in Figure 11, and the results obtained are shown in Figures 12 and 13. When the lightning stroke strikes the transmission line and the voltage at the first insulator reaches design voltage, flashover occurs. The voltage waveform is transmitted to the second insulator. However, remark that the current absorbed by the second insulator is negligible since the voltage at that point is lower than the design one, thus no flashover occurs.

3.3. LIGHTNING ARRESTER.

This example is similar to the first one, but in this case the second insulator has been replaced by a lightning arrester, which has 650 KV voltage reference instead of 1.6 MegV that has the insulator. The value of I_0 has been decreased to 12 KA so the insulator flashover does not occur. However, the voltage peak is high enough to cause the lightning arrester to sparkover.

```
.SUBCKT ARRESTER 2 10
**** The lines that differs from the
**** INSULATOR model are listed above
R10 2 3 20
R21 7 8 100.78K
E14 11 0 10 0 0.3912024
**** The rest of the model is the same
**** than the subcircuit INSULATO
```

Figure 14: Extract of the lightning arrester model



Figure 15: Voltage at point of stroke, insulator and lightning arrester.



Figure 16: Current absorbed by insulator and lightning arrester.

In Figure 15, the voltage waveform is transmitted beyond the insulator through the line since the voltage peak is lower than his reference. Therefore, no current flows through it (see Figure 16). In contrast, the lightning arrester sparks over according to his lower voltage reference and the current is derived to ground.

CONCLUSIONS

In the model of insulator that we have designed, the parameters (like delay time, internal resistance, sparkover voltage, ...) could be changed, simulating as a real lightning arrester as well as an ideal case.

This model could run in SPICE and in PSPICE to simulate the behaviour of an insulator, lightning arrester and other electrical devices which have non-linear comportment.

REFERENCES

/1/ PSPICE: Microsim Corp., Laguna Hills. CA. January 1989.

/2/ L. A. KRAFT: "Modelling lightning perfomance of transmission systems using PSPICE", IEEE Transactions on Power Systems, Vol. 6,No. 2, May 1991.

/3/ SPICE: "An adaption of UC Berkeley, Vers. 1.41", Intusoft 1987.

/4/ L. HAYDOCK: "Application of a Finite Element Technique to Model an Electrical Power Cable", IEEE Transactions on Magnetics, Vol. 30, No. 5, September 1994.

/5/ J. A. DIAS PINTO, A. P. BREDA COIMBRA AND C. F. LEMOS ANTUNES: "Finite element electro-thermal approach of EHV power cables", 10th conference on the computation of electromagnetic fields, pag. 374, Berlin, Germany, 1995.

/6/ M. HASLER ET J. NEIRYNCK: "Circuits non linéaires", Presses polytechniques romandes, Lausanne 1985.